SISTEMI MECCANICI E MODELLI

Homework 1 (testo 1): ANALISI DI UN MECCANISMO A GINOCCHIERA

Andrea Cardone
185096
andrea.cardone-1@studenti.unitn.it

Si consideri il meccanismo della figura. Le dimensioni x0 e y0 sono tali che quando le aste L sono allineate in verticale il meccanismo è nella configurazioni di ginocchiera chiusa (si trascuri lo spessore del pattino), con le aste r e L1 allineate e y = 0.

Assumendo L=1, L1=L 19/10 ed r= L/10 determinare:

1) Analisi di posizione. Determinare la corsa y della pressa in funzione di q. Rappresentare graficamente y per q ∈ [q0-π/4, q0+π/4]
essendo q0 l'angolo movente che corrisponde alla ginocchiera chiusa.

2) Analisi di velocità. Determiare il rapporto di velocità τyq. (Suggerimento, sfruttare il fatto che le catene cinematiche sono debol-
mente accoppiate). Rappresentare graficamente τyq per q ∈ [q0-π/4, q0+π/4].

3) Analisi di accelerazione (Facoltativo). Assumendo q=ω t, con ω=1, Determinare la accelerazione y... Rappresentare graficamente y.. per
q ∈ [q0-π/4, q0+π/4].

4) Animazione. Rappresentare graficamente il meccanismo in funzione di q, per q ∈ [q0-π, q0+π]

5) Configurazioni singolari. Determinare le configurazioni singolari.



1) Analisi di posizione

Consideriamo il meccanismo a ginocchiera, con un grado di libertà e procediamo con la scrittura delle equazioni di chiusura.

```
eq₁ = L₁ Cos[q[t]] + L₂ Cos[θ₂[t]] - L₃ Cos[θ₃[t]] - L₆ == 0;
eq₂ = L₁ Sin[q[t]] + L₂ Sin[θ₂[t]] - L₃ Sin[θ₃[t]] == 0;
eq₃ = L₄ Cos[θ₄[t]] + L₃ Cos[θ₃[t]] == 0;
eq₄ = L₄ Sin[θ₄[t]] + L₅[t] + L₃ Sin[θ₃[t]] == 0;
```

Per l'analisi di posizione conviene sfruttare il fatto che il meccanismo a ginocchiera è debolmente accoppiato, come dimostra la matrice Jacobiana calcolata in seguito. Risulta quindi più agevole risolvere il prima il quadrilatero e in seguito la parte rimanente composta da una biella e un pattino.

```
f₁ = First@eq₁;
f₂ = First@eq₂;
f₃ = First@eq₃;
f₄ = First@eq₄;
J = D[{f₁, f₂, f₃, f₄}, {{θ₂[t], θ₃[t], θ₄[t], L₅[t]}}];
TraditionalForm[MatrixForm[J]]
```

$$\begin{pmatrix} L_2\,(-\sin(\theta_2(t))) & L_3 \sin(\theta_3(t)) & 0 & 0 \\ L_2 \cos(\theta_2(t)) & L_3\,(-\cos(\theta_3(t))) & 0 & 0 \\ 0 & L_3\,(-\sin(\theta_3(t))) & L_4\,(-\sin(\theta_4(t))) & 0 \\ 0 & L_3 \cos(\theta_3(t)) & L_4 \cos(\theta_4(t)) & 1 \end{pmatrix}$$

Per risolvere il quadrilatero è conveniente sfruttare alcune considerazioni geometriche per rendrere più agevole il calcolo delle soluzioni in forma chiusa, senza dover risolvere il sistema delle equazioni di chiusura.



```
MGinocchiera[q_, x0_, y0_, r_, L1_, L_, modo_, xD_, yD_] := Module[
  {xB, yB, L5, cα, α, θ5, θ2, xC, yC, θ3, xE, yE},
  xB = x0 + r Cos[q];
  yB = y0 + r Sin[q];
  L5 = Sqrt[(xD - xB)^2 + (yD - yB)^2];
  cα = (L5^2 + L1^2 - L^2)/(2 L5 L1);
  α = ArcCos[cα];
  θ5 = ArcTan[xD - xB, yD - yB];
  If[modo ≥ 0, θ2 = θ5 + α, θ2 = θ5 - α];
  xC = xB + L1 Cos[θ2];
  yC = yB + L1 Sin[θ2];
  θ3 = ArcTan[xC - xD, yC - yD];
  xE = xD;
  yE = 2 yC - yD;
  {θ2, θ3, {{x0, y0}, {xB, yB}, {xC, yC}, {xD, yD}}, {xE, yE}}
 ]
```

In seguto si calcolano l’angolo q_0, corrispondente alla configurazione di ginocchiera chiusa, e y(q), funzione della posizione del pattino, con i dati del meccanismo assegnato, scegliendo il modo di assemblaggio corrispondente.

```
q0 = -ArcTan[0.1 + 1.9, 1];
y[q_] := Simplify[MGinocchiera[q, 0, 2, 0.1, 1.9, 1, -1, Sqrt[3], 2]][[4, 2]]
TraditionalForm[y[q]]
```

$$y(q) = -3.7999999999999994\grave{}\ \sin\left(\cos^{-1}\left(\frac{1.4789473684210523\grave{} - 0.09116056881941458\grave{}\ \cos(q)}{\sqrt{3.01\grave{} - 0.34641016151377546\grave{}\ \cos(q)}}\right) - \tan^{-1}(1.7320508075688772\grave{} - 0.1\grave{}\ \cos(q), -0.1\grave{}\ \sin(q))\right) + 0.2\grave{}\ \sin(q) + 2.\grave{}$$

```
Plot[y[q], {q, q0 - π/4, q0 + π/4}]
```



2) Analisi di velocità

Per calcolare τ_{yq} è sufficiente derivare y(q(t)), calcolata al punto precedente, rispetto a t e in seguito calcolare il rapporto di velocità $\tau_{yq} = \frac{y'(q(t))}{q'(t)}$

```
vy = Simplify[∂t y[q[t]]];
vq = ∂t q[t];
τyq[q_] := Evaluate[vy/vq /. q[t] → q]
TraditionalForm[τyq[q]]
```

$$\tau_{yq}(q0) = .2` \cos(q) - \frac{1}{1.` \cos(q) - 8.689121551303865`} 3.8` \left((\sin(q)(0.05263157894736845` - 0.045580284409707295` \cos(q)))\Big/\left(\sqrt{3.01` - 0.34641016151377546` \cos(q)}\ \sqrt{\left((0.023989623373530153` (16.22354256422848` - 1.` \cos(q))^2)\big/(1.` \cos(q) - 8.689121551303867`) + 1\right)}\right) + 0.02886751345948129` \sin^2(q) + 0.02886751345948129` \cos(q)(1.` \cos(q) - 17.32050807568877`)\right) \cos\left(\cos^{-1}\left(\frac{1.4789473684210523` - 0.09116056881941459` \cos(q)}{\sqrt{3.01` - 0.34641016151377546` \cos(q)}}\right) - \tan^{-1}(1.7320508075688772` - 0.1` \cos(q), -0.1` \sin(q))\right)$$

```
Plot[τyq[q], {q, q0 - π/4, q0 + π/4}]
```



3) Analisi accelerazione

Per ricavare l'accelerazione del pattino è sufficiente derivare due volte y(q(t)) due volte rispetto al tempo, assumendo q(t) = ωt con ω = 1

```
a_y[q_] := Evaluate[Simplify[∂_t ∂_t y[q[t]]] /. q[t] → q /. q′[t] → 1]
Plot[a_y[q], {q, q0 - π/4, q0 + π/4}]
```



4) Animazione

```
mghw1[q_] := Evaluate[Simplify[MGinocchiera[q, 0, 2, 0.1, 1.9, 1, -1, √3, 2]]]
```

```
Manipulate[Show[
  ListLinePlot[{mghw1[q][[3, 1]], mghw1[q][[3, 2]]},
   PlotRange → {{-0.5, 2.5}, {-0.5, 2.5}}, PlotStyle → Thick,
   AspectRatio → 1, PlotMarkers → {"○"}],
  ListLinePlot[{mghw1[q][[3, 2]], mghw1[q][[3, 3]]},
   PlotMarkers → {"○"}, PlotStyle → Thick],
  ListLinePlot[{mghw1[q][[3, 3]], mghw1[q][[3, 4]]},
   PlotMarkers → {"○"}, PlotStyle → Thick],
  ListLinePlot[{mghw1[q][[3, 3]], mghw1[q][[4]]},
   PlotMarkers → "○", PlotStyle → Thick],
  ListLinePlot[{{√3, 3}, {√3, -3}}, PlotStyle → {Dashed, Red, Thin}],
  ListPlot[{{0, 2}, {√3, 2}}, PlotMarkers → {"⊕"}],
  ListPlot[{mghw1[q][[4]]}, PlotMarkers → {"□", Large}]],
 {q, q0 - π, q0 + π}]
```



5) Configurazioni singolari

Le configurazioni singolari possono essere determinate trovando la configurazione che annulla il determinante della matirce Jacobiana. In seguito viene risolto Det(J) in funzione della variabile $\theta_2(t)$

```
TraditionalForm[Simplify[Solve[Det[J] == 0, {θ2[t]}] /. C[1] → 0]]
```

$$\left\{\left\{\theta_2(t) \to \tan^{-1}(-\cos(\theta_3(t)), -\sin(\theta_3(t)))\right\}, \left\{\theta_2(t) \to \tan^{-1}(\cos(\theta_3(t)), \sin(\theta_3(t)))\right\}\right\}$$

Da questa soluzione si deduce che le configurazioni singolari del meccanismo a ginocchiera sono le medesime del quadrilatero articolato, cioè:

$\theta_2(t) = \theta_3(t)$

$\theta_2(t) = \theta_3(t) + \pi$

Non resta che verificare se queste uguaglianze hanno soluzione nel meccanismo a ginocchiera considerato.

```
θ2[q_] := Evaluate[Simplify[MGinocchiera[q, 0, 2, 0.1, 1.9, 1, -1, √3, 2]][[1]]]
θ3[q_] := Evaluate[Simplify[MGinocchiera[q, 0, 2, 0.1, 1.9, 1, -1, √3, 2]][[2]]]
Plot[{θ2[q], θ3[q], θ3[q] + π}, {q, 0, 2 π},
 PlotLegends → {"θ2[q]", "θ3[q]", "θ3[q]+π"}]
```



Dall’analisi grafica si deduce che il meccanismo considerato non ha configurazioni singolari.